


تجربة التحكم بدرجة الحرارة

المبدلات ADC,DAC


تجربة التحكم بدرجة الحرارة:

تتضمن دارة التجارب لدينا على سخان حراري ممثل بمقاومة ذات استطاعة 2.5 واط ويمكن التحكم بتشغيلها وإطفائها من خلال البت السابع في البوابة A . ترتبط حالة السخان مع الليد الضوئي المتصل إلى جوار السخان الحراري وحساس حراري موجود بالقرب من المقاومة أيضاً ويقوم بقياس الحرارة عبر مضخم عمليات متصل مع المقاومة المتغيرة VR3 . يتم تمرير خرج هذا المكبر إلى مقارن عبر مفاتيح التحكم SW1,SW2 . ومن ثم يتم تحويل القيمة التماثلية للحرارة إلى قيمة رقمية باستخدام دارة ADC .

نقوم بهذه التجربة بتشغيل السخان ومن ثم القيام بعملية التحكم بدرجة الحرارة وعند الوصول إلى حرارة محددة يتم القيام بعملية تدوير المحرك وبالتالي يتم تبريد السخان. عند انخفاض الحرارة تحت القيمة المرجعية يتم عندها إيقاف المحرك عن الدوران.



المخطط التدفقي لعمل نظام التحكم بدرجة الحرارة:



```

PPIAA:    EQU    00H        ;U10 8255A PORT A
PPIAB:    EQU    02H        ;PORT B
PPIAC:    EQU    04H        ;PORT C
PPIAK:    EQU    06H        ;CONTROL
COMP:     EQU    10H        ;COMPARITOR BIT 4
DELTIME:  EQU    1000H     ;SET DELAY TIME
ADCTIME   EQU    10H        ;DAC SETTling TIME
SETPNT    EQU    60H        ;TEMP SET POINT

                ORIGIN      0050H:0100H
START:      MOV    AL,80H    ;SET MOTOR OFF, HEATER ON
            OUT    PPIAK,AL
            MOV    DL,AL     ;SAVE OUTPUT
DOAGAIN:    MOV    AL,80H    ;SET CONTROL BYTE FOR PPI
            OUT    PPIAK,AL  ;OUTPUT TO CONTROL REG
            MOV    AL,DL     ;GET LAST OUTPUT
            OUT    PPIAA,AL  ;AND OUTPUT
            CALL   DELAY     ;WAIT
            CALL   ATOD     ;READ A TO D
            MOV    AL,CL     ;GET TEMP
            SUB    AL,SETPNT ;COMPARE SET POINT
            JAE   MOTORON   ;START MOTOR TO COOL HEATER
            MOV    AL,80H    ;ELSE TURN MOTOR OFF
            MOV    DL,AL     ;SAVE OUTPUT
            JMP    DOAGAIN   ;DO FOREVER
MOTORON:    MOV    AL,0C0H   ;MOTOR ON TO COOL
            MOV    DL,AL     ;SAVE OUTPUT
            OUT    PPIAA,AL  ;AND OUTPUT
            JMP    DOAGAIN   ;DO FOREVER

;*****
;      A TO D ROUTINE
;      RESULT IN CL
ATOD:      MOV    AL,90h     ;SET CONTROL BYTE FOR PPI
            OUT    PPIAK,AL  ;OUTPUT TO CONTROL REG
            MOV    CL,00     ;SET OUT RAMP TO ZERO
REPEAT:    MOV    AL,CL     ;STORE FOR OUTPUT
            OUT    PPIAC,AL  ;OUTPUT TO DAC
            INC    CL        ;INCREMENT RAMP
            CALL   ADCDELAY   ;ALLOW TIME TO SETTLE
            IN     AL,PPIAA   ;READ PORT A
            TEST   AL,COMP    ;TEST COMPARITOR OUTPUT
            JNZ   CONVCOMP   ;IF NOT ZERO, EOC REPEAT CYCLE
            JMP    REPEAT    ;ELSE CONTINUE RAMPING

CONVCOMP:  MOV    AL,80H     ;SET CONTROL BYTE FOR PPI
            OUT    PPIAK,AL  ;OUTPUT TO CONTROL REG
            RET

```

```

;*****
; DAC DELAY SUBROUTINE

ADCDELAY:  PUSH  CX          ;RESERVE REG
           MOV   CX,ADCTIME ;SET DELAY COUNTER
NEXTADC:  LOOP  NEXTADC     ;COUNT = COUNT-1, LOOP IF NOT 0
           POP   CX          ;RESTORE REG
           RET

;*****
; DELAY SUBROUTINE



DELAY:     PUSH  CX          ;RESERVE REG
           MOV   CX,DELTIME ;SET DELAY COUNTER
NEXT:     LOOP  NEXT        ;COUNT = COUNT-1, LOOP IF NOT 0
           POP   CX          ;RESTORE REG
           RET

END

```

ميدلات DAC,ADC:

يتم توصيل المبدل ذو البتات الثمانية إلى البوابة 2 ويتم التحكم بها من خلال البوابة B. هنا يتم إخراج إشارة تماثلية من اللوحة من أجل التحكم بمكبر الصوت الموجود على اللوحة. إن خرج المبدل الرقمي التماثلي يقابل 2.5 فولت من أجل 255 مستوى (أو تقابل القيمة FF hex) ويمكن تخميد الخرج باستخدام المقاومة المتغيرة VR1 في قسم DAC & COMPARATOR. سيتم توليد موجة سن منشار في هذه التجربة يتم ضبط ترددها من خلال التأخير الزمني ويتم ضبط مفاتيح التحكم كما في الشكل التالي:



```

PPIAA: EQU 00H ;U10 8255A PORT A
PPIAB: EQU 02H ;PORT B
PPIAC: EQU 04H ;PORT C
PPIAK: EQU 06H ;CONTROL

DELTIME: EQU 10H ;DELAY COUNT

ORIGIN 0050H:0100H
START: MOV AL,90h ;SET CONTROL BYTE FOR PPI
      OUT PPIAK,AL ;OUTPUT TO CONTROL REG
      MOV AL,00 ;SET OUT COUNT TO ZERO
REPEAT: OUT PPIAC,AL ;OUTPUT TO PORT C (LED's)
      CALL DELAY ;DELAY
      INC AL ;INCREMENT OUTPUT COUNT
      JMP REPEAT ;DO FOREVER



DELAY: MOV CX,DELTIME ;SET DELAY COUNTER
NEXT: LOOP NEXT ;COUNT = COUNT-1, LOOP IF NOT 0
      RET

END

```

التبديل التماثلي الرقمي ADC:

تتم عملية التبديل ضمن اللوحة من قبل المبدل الرقمي التماثلي بالإضافة إلى المقارن والذي سنقوم بالتحكم به برمجياً , الطريقة الأسهل هي وصل أحد مداخل المقارن إلى الجهد المراد قياسه والآخر إلى خرج المبدل التماثلي الرقمي وبالتالي يمكن تحديد قيمة إشارة الخرج إما صفر أو واحد حسب ناتج عملية المقارنة.



```

PPIAA: EQU 00H ;U10 8255A PORT A
PPIAB: EQU 02H ;PORT B
PPIAC: EQU 04H ;PORT C
PPIAK: EQU 06H ;CONTROL
COMP: EQU 10H ;COMPARITOR BIT 4

DELTIME EQU 10H ;DELAY COUN

ORIGIN 0050H:0100H
START: MOV AL,90h ;SET CONTROL BYTE FOR PPI
OUT PPIAK,AL ;OUTPUT TO CONTROL REG
CYCLE: MOV CL,00 ;SET OUT RAMP TO ZERO
REPEAT: MOV AL,CL ;STORE FOR OUTPUT
OUT PPIAC,AL ;OUTPUT TO DAC
INC CL ;INCREMENT RAMP
CALL DELAY ;ALLOW TIME TO SETTLE
IN AL,PPIAA ;READ PORT A
TEST AL,COMP ;TEST COMPARITOR OUTPUT
JNZ CYCLE ;IF NOT ZERO, EOC REPEAT CYCLE
JMP REPEAT ;ELSE CONTINUE RAMPING
;*****
; DELAY SUBROUTINE
DELAY: PUSH CX ;RESERVE REG
MOV CX,DELTIME ;SET DELAY COUNTER
NEXT: LOOP NEXT ;COUNT = COUNT-1, LOOP IF NOT
0
POP CX ;RESTORE REG
RET
END

```